
8 Flow through a Jointed Rock Slope

8.1 Problem Statement

The stability of a slope in jointed rock is affected by the water level behind the slope. In this
example, the water level is raised in stages until the slope becomes unstable. The failure of the
slope occurs when the fluid pressure in the joints increases (and the effective normal stress in the
joints decreases) such that the limiting shear strength of the joints at the slope face is reached.

The problem geometry, shown in Figure 8.1, consists of a slope in regularly jointed rock. The
water level is raised in four stages to elevations of 6 m, 8 m, 9 m and 10 m above the slope toe. A
steady-state flow analysis is performed at each stage.

Figure 8.1 Problem geometry for example problem involving flow through
jointed rock slope

The following material properties are assumed for the jointed rock slope.

Rock Properties:

density 2500 kg/m3

bulk modulus 16.7 GPa
shear modulus 10.0 GPa

Joint Mechanical Properties:

normal stiffness 10 GPa/m
shear stiffness 10 GPa/m
friction angle 25◦

Joint Hydraulic Properties:

permeability factor 1 × 108 MPa−1 sec−1

residual hydraulic
aperture 2 × 10−4 m
aperture at zero
normal stress 5 × 10−4 m

Fluid Properties:

density 1000 kg/m3

8.2 UDEC Analysis

The UDEC model is shown in Figure 8.2. The problem is modeled as a steady-state flow analysis by
specifying block fluid steady-state. The water level is raised by changing the fluid pressure gradient
for each stage with the block edge apply pore-pressure pressure-gradient-y command. The data file
is listed in Section 8.3.

Initially, the slope is brought to an equilibrium state under gravity loading. Then, the water level
at the right-hand side is raised to 6 m above the slope toe; the water level on the left-hand side is
maintained at the level of the slope toe. The slope is stable for this fluid pressure condition. The
steady-state flow pattern for this condition is shown in Figure 8.3.

Next, the right-hand water level is raised to 8 m. The steady-state flow condition for the 8-m water
height is shown in Figure 8.4. Again, the system is in equilibrium. The water level is then raised
to 9 m. The flow condition is shown in Figure 8.5; the slope is still stable.

Finally, the water level is raised to the top of the slope. The flow pattern for this case is shown in
Figure 8.6. With the water level at 10 m, the slope slides, as indicated by the displaced rock wedge
shown in Figure 8.7.

Figure 8.2 Problem geometry for example problem involving flow through
jointed rock slope

Figure 8.3 Water level at right-hand side equal to 6 m (maximum flow rate
= 5.6 × 10−5 m3/sec)

Figure 8.4 Water level at right-hand side equal to 8 m (maximum flow rate
= 7.4 × 10−5 m3/sec)

Figure 8.5 Water level at right-hand side equal to 9 m (maximum flow rate
= 8.1 × 10−5 m3/sec)

Figure 8.6 Water level at right-hand side equal to 10 m (maximum flow rate
= 9.4 × 10−5 m3/sec)

Figure 8.7 Slope failure with water level at 10 m

8.3 Listing of Data File

Example 8.1 SLOPEFLO.DAT

model new
mdel title ’Flow Through a Jointed Rock Slope’
; ================================================
;
; --- fluid flow test run ---
;
; --- slope : 10 m high ---
;
; --- 2 joint sets : 20 and 80 deg.
; friction = 30 deg.
;
; --- r.h.s. water level : 6 m --- no failure ---
; 8 m
; 9 m
; 10 m --- failure ---
;
; ================================================
block config fluid
block fluid clear steady-state off
block tolerance corner-round-length 0.05
block create polygon 0 -5 0 0 5 0 11 10 23 10 23 -5
block cut joint-set angle 20 spacing 2 origin 5 1
block cut joint-set angle 80 spacing 3 origin 5 0
block delete range area 0.1
block zone gen edge 10.0
block zone group ’rock’
block zone cmodel assign elastic density 2.5E-3 bulk 1.66667E4 ...

shear 1E4 range group ’rock’
block contact group ’joint’
block contact cmodel assign area stiffness-shear 1E4 ...

stiffness-normal 1E4 friction 45 permeability-factor 1E8 ...
aperture-residual 0.0002 aperture-zero-load 0.0005 range group ’joint’

; new contact default
block contact cmodel default area stiffness-shear 1E4 ...

stiffness-normal 1E4 friction 45 permeability-factor 1E8 ...
aperture-residual 2E-4 aperture-zero-load 5E-4

block fluid property density 0.0010
block insitu stress -0.125 0.0 -0.25 ...

gradient-x 0.0 0.0 0.0 gradient-y 0.0125 0.0 0.025
block gridpoint apply velocity-x 0 range pos-x -0.1 0.1 pos-y -5.1 0.1
block gridpoint apply velocity-x 0 range pos-x 22.5 23.1 pos-y -5.1 10.1
block gridpoint apply velocity-y 0 range pos-x -0.1 23.1 pos-y -5.1 -4.9

model gravity 0 -10
block gridpoint history displacement-x 11.0 10.0
block gridpoint history displacement-y 11.0 10.0
block mechanical history unbalanced-maximum
block solve ratio 1.0E-5 elastic
model save ’slfl1.sav’
;
; --------------------------------------
;
; flow --- r.h.s. water at y=6m
;
; --- no failure ---
;
; --------------------------------------
;
block fluid steady-state on
block edge apply impermeable range pos-x -0.1 23.1 pos-y -5.1 -4.9
block edge apply pore-pressure 0.0 pressure-gradient-y -0.01 ...

range pos-x -0.1 0.1 pos-y -5.1 0.1
block edge apply pore-pressure 0.06 pressure-gradient-y -0.01 ...

range pos-x 22.5 23.1 pos-y -5.1 6
block gridpoint history displacement-x 5.92 1.54
block gridpoint history displacement-y 5.92 1.54
block contact group ’weak joint’
;group joint ’weak joint’ range group ’joint’ ; changed name
block contact cmodel assign area stiffness-shear 1E4 ...

stiffness-normal 1E4 friction 25 permeability-factor 1E8 ...
nonwetting-permeability-factor 1 aperture-residual 0.0002 ...
aperture-zero-load 0.0005 constant-b 1 exponent 3 ...
range group ’weak joint’

; set new contact default
block contact cmodel default area stiffness-shear 1E4 ...

stiffness-normal 1E4 friction 25 permeability-factor 1E8 ...
nonwetting-permeability-factor 1 aperture-residual 0.0002 ...
aperture-zero-load 0.0005 constant-b 1 exponent 3

block solve ratio 1.0E-5
model save ’slfl2.sav’
;
; --------------------------------------
;
; flow --- r.h.s. water raised to y=8m
;
; --- some slip, but no block failure ---
;
; --------------------------------------
;

block edge apply pore-pressure 0.08 ...
pressure-gradient-y -0.01 range pos-x 22.5 23.1 pos-y -5.1 8

block solve ratio 1.0E-5
model save ’slfl3.sav’
;
; --------------------------------------
;
; flow --- r.h.s. water raised to y=9m
;
; --- no failure ---
;
; --------------------------------------
;
block edge apply pore-pressure 0.09 ...

pressure-gradient-y -0.01 range pos-x 22.5 23.1 pos-y -5.1 9.1
block solve ratio 1.0E-5
model save ’slfl4.sav’
;
; --------------------------------------
;
; flow --- r.h.s. water raised to y=10m
;
; --- failure ---
;
; --------------------------------------
;
block edge apply pore-pressure 0.10 ...

pressure-gradient-y -0.01 range pos-x 22.5 23.1 pos-y -5.1 10.1
block cycle 15000
model save ’slfl5.sav’
return

